計算機アルゴリズム特論

微分方程式の数値解法について

微分方程式

$$\frac{\mathrm{d}u}{\mathrm{d}x} = f(x,u), \qquad u(0) = a \tag{1}$$

を解くための数値解法は数多く知られている。そのなかでもっとも簡単なものをいくつか紹介する。

通常，数値解法では x 軸上のすべての点 x での解 $u(x)$ を求めるのでなく，ある微小な間隔 (ここでは Δx とする）ごとに近似解を求めていく。以下では，初期点 (ここでは 0) を x_0 とし n 番目の点 x_n とする。すなわち，

$$x_n = x_0 + n\,\Delta x, \qquad n = 0,1,2,\ldots$$

とする。また，x_n における近似解を u_n とする。

- オイラー法

$$\begin{aligned} u_{n+1} &= u_n + \Delta x\, f(x_n, u_n), \qquad n = 0,1,2,\ldots \\ u_0 &= a \end{aligned} \tag{2}$$

- ホイン法

$$\begin{aligned} u_{n+1} &= u_n + \frac{\Delta x}{2}\Big(f(x_n,u_n) + f(x_{n+1}, u_n + \Delta x f(x_n,u_n))\Big), \qquad n = 0,1,2,\ldots \\ u_0 &= a \end{aligned} \tag{3}$$

- 後退オイラー法

$$\begin{aligned} u_{n+1} &= u_n + \Delta x\, f(x_{n+1}, u_{n+1}), \quad n = 0,1,2,\ldots, \\ u_0 &= a \end{aligned} \tag{4}$$

  後退オイラー法では，現時点では未知な値 u_{n+1} が右辺の関数 $f(x,u)$ の中に入っているので，1-step 前進するたびに非線形方程式

$$F(u) := u - u_n - \Delta x\, f(x_{n+1}, u) = 0$$

  を u について解き，得られた値を u_{n+1} とする。

- 台形公式

$$\begin{aligned} u_{n+1} &= u_n + \frac{\Delta x}{2}\Big(f(x_n,u_n) + f(x_{n+1}, u_{n+1})\Big), \qquad n = 0,1,2,\ldots \\ u_0 &= a \end{aligned} \tag{5}$$

  台形公式でも 1 step 前進するごとに方程式

$$F(u) := u - u_n - \frac{\Delta x}{2}\Big(f(x_n,u_n) + f(x_{n+1}, u)\Big) = 0$$

  を解かねばならない。

非線形方程式を解くための代表的な数値解法にニュートン法がある。ニュートン法とは，方程式

$$F(u) = 0$$

の解を適当な初期値 $u^{[0]}$ を与えて漸化式

$$u^{[k+1]} = u^{[k]} - \frac{F(u^{[k]})}{F'(u^{[k]})}, \qquad k = 0, 1, \ldots$$

で解く方法である。

後退オイラー法の数値例

微分方程式

$$\frac{\mathrm{d}u}{\mathrm{d}x} = \exp(-u), \quad u(0) = 0$$

を後退オイラー法

$$u_{n+1} = u_n + \Delta x \, \exp(-u_{n+1}), \qquad u_0 = 0, \qquad n = 0, 1, 2, \ldots,$$

で解く。この方程式の真の解は

$$u(x) = \log(x+1)$$

である。刻み幅 Δx を $\Delta x = 1/2^i \,(i = 3, 4, \ldots, 12)$ と変えて行き，解の $x = 1$ における値 $\log 2 = 6.9314718055994530941 \cdots \times 10^{-1}$ と $x = 1 \,(= 2^i \,\Delta x)$ における数値解 u_{2^i} との差，すなわち誤差を調べたものを下の表に示す。

i	u_{2^i}	$\log_2$｜誤差｜
3	6.7251548E-01	-5.599E+00
4	6.8258180E-01	-6.565E+00
5	6.8779925E-01	-7.547E+00
6	6.9045652E-01	-8.538E+00
7	6.9179763E-01	-9.533E+00
8	6.9247134E-01	-1.053E+01
9	6.9280900E-01	-1.153E+01
10	6.9297802E-01	-1.253E+01
11	6.9306258E-01	-1.353E+01
12	6.9310488E-01	-1.453E+01

Fortran 90 のプログラムを下に示す。

```
!-----------------------------------------------------------
!     Backward Euler method
!
!     Solving u'=exp(-u),  u(0)=0,
!
```

```
!     Exact solution is
!      u(x)=log (x+1)
!-------------------------------------------------------------
Program Backward_Euler
  implicit none
  integer:: i
  real (8):: dx,u0,u1,x0,x1,T,f

  T=log(2.d0)
  write (*,'("----------------------------------------")')
  write (*,'(" log_2 dx      u         log_2(|error|)     ")')
  write (*,'("----------------------------------------")')

  do i=3,12
     dx=2.d0**(-i)
     u0=0.d0
     x0=0.d0

     do while (x0+dx <=1.d0)
        call solve(u0,dx,u1)
        u1=u0+dx*f(u1)
        x1=x0+dx

        x0=x1
        u0=u1
     end do
     write (*,'("  ",i3,"   ",1pd15.7,"  ",1pd10.3)') -i,u0,log2(abs(u0-T))
  end do
  write (*,'("----------------------------------------")')

end Program Backward_Euler

!-------------------------------------------------------------
! Solving u_1-h f(u_1)-u_0=0 for u_1 by Newton's method
!-------------------------------------------------------------
subroutine solve(u0,dx,u1)
  implicit none
  real (8):: y0,y1,u0,u1,dx,err,f,g

  y0=u0
  err=1.d10
  do while (err > 1.d-14)
     y1=y0-(y0-u0-dx*f(y0))/(1.d0-dx*g(y0))
     err=abs(y1-y0)/abs(y1)
     y0=y1
  end do

  u1=y0
  return
end subroutine solve

function f(x)
  implicit none
  real (8):: f,x
  f=exp(-x)
  return
end function f

function g(x)   !  g(x)=f'(x)
  implicit none
  real (8):: g,x
  g=-exp(-x)
  return
end function g
```

前進 Euler 法と後退 Euler 法の比較

微分方程式

$$\frac{\mathrm{d}u}{\mathrm{d}x} = -x\,u, \quad u(0) = 1$$

を Euler 法と後退 Euler 法によって解き，結果を比較する。刻み幅はどちらも $\Delta x = 0.3$ とする。この方程式の真の解は $u(x) = \exp\left(-x^2/2\right)$ である。



前進 Euler 法で求めた解



後退 Euler 法で求めた解

問題 1 $y'(x) = 2\,y(x)(1 - y(x))$, $y(0) = 1/2$ を $\Delta x = 1/2^i\,(i = 1, 2, \ldots, 5)$ と変化させながら前進 Euler 法で解き $x = 1$ における誤差を求めよ。ただし，真の解は $y(x) = \exp(2x)/(\exp(2x) + 1)$ である。

問題 2 同じことをホイン法で行え。

問題 3 同じことを後退 Euler 法で行え。

http://www.pna.eis.akita-pu.ac.jp/~ozawa/Lecture/Grad_Schl/HandOut02.pdf